

NAG Library Routine Document

G05SQF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

G05SQF generates a vector of pseudorandom numbers uniformly distributed over the interval $[a, b]$.

2 Specification

```
SUBROUTINE G05SQF(N, A, B, STATE, X, IFAIL)
INTEGER          N, STATE(lstate), IFAIL
double precision A, B, X(N)
```

3 Description

If $a = 0$ and $b = 1$, G05SQF returns the next n values y_i from a uniform $(0, 1]$ generator (see G05SAF for details).

For other values of a and b , G05SQF applies the transformation

$$x_i = a + (b - a)y_i.$$

The routine ensures that the values x_i lie in the closed interval $[a, b]$.

One of the initialization routines G05KFF (for a repeatable sequence if computed sequentially) or G05KGF (for a non-repeatable sequence) must be called prior to the first call to G05SQF.

4 References

Knuth D E (1981) *The Art of Computer Programming (Volume 2)* (2nd Edition) Addison–Wesley

5 Parameters

- | | | |
|----|---|----------------------------|
| 1: | N – INTEGER | <i>Input</i> |
| | <i>On entry:</i> n , the number of pseudorandom numbers to be generated. | |
| | <i>Constraint:</i> $N \geq 0$. | |
| 2: | A – <i>double precision</i> | <i>Input</i> |
| 3: | B – <i>double precision</i> | <i>Input</i> |
| | <i>On entry:</i> the end points a and b of the uniform distribution. | |
| | <i>Constraint:</i> $A \leq B$. | |
| 4: | STATE(<i>lstate</i>) – INTEGER array | <i>Communication Array</i> |
| | Note: <i>lstate</i> = LSTATE, the dimension of STATE as supplied to the initialization routine G05KFF or G05KGF. | |
| | <i>On entry:</i> contains information on the selected base generator and its current state. | |
| | <i>On exit:</i> contains updated information on the state of the generator. | |
| 5: | X(N) – <i>double precision</i> array | <i>Output</i> |
| | <i>On exit:</i> the n pseudorandom numbers from the specified uniform distribution. | |

6: IFAIL – INTEGER

Input/Output

On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

On exit: IFAIL = 0 unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1 , explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

On entry, $N < 0$.

IFAIL = 3

On entry, $B < A$.

IFAIL = 4

On entry, STATE vector was not initialized or has been corrupted.

7 Accuracy

Not applicable.

8 Further Comments

Although y_i takes a value from the half closed interval $(0, 1]$ and $x_i = a + (b - a)y_i$, x_i is documented as taking values from the closed interval $[a, b]$. This is because for some values of a and b , G05SQF may return a value of a due to numerical rounding.

9 Example

This example prints five pseudorandom numbers from a uniform distribution between -1.0 and 1.0 , generated by a single call to G05SQF, after initialization by G05KFF.

9.1 Program Text

```
*      G05SQF Example Program Text
*      Mark 22 Release. NAG Copyright 2006.
*      .. Parameters ..
      INTEGER                NOUT
      PARAMETER              (NOUT=6)
      INTEGER                MSTATE, MSEED, N
      PARAMETER              (MSTATE=633, MSEED=1, N=5)
*      .. Local Scalars ..
      DOUBLE PRECISION       A, B
      INTEGER                GENID, I, IFAIL, LSEED, LSTATE, SUBID
*      .. Local Arrays ..
      DOUBLE PRECISION       X(N)
      INTEGER                SEED(MSEED), STATE(MSTATE)
*      .. External Subroutines ..
      EXTERNAL               G05KFF, G05SQF
```

```

*      .. Executable Statements ..
      WRITE (NOUT,*) 'G05SQF Example Program Results'
      WRITE (NOUT,*)
*      Set the distribution parameters A and B
      A = -1.0D0
      B = 1.0D0
*      Initialize the seed to a repeatable sequence
      SEED(1) = 1762543
*      GENID and SUBID identify the base generator.
      GENID = 1
      SUBID = 1
*      Initialize the generator to a repeatable sequence
      LSTATE = MSTATE
      LSEED = MSEED
      IFAIL = 1
      CALL G05KFF(GENID,SUBID,SEED,LSEED,STATE,LSTATE,IFAIL)
      IF (IFAIL.NE.0) THEN
         WRITE (NOUT,99998) IFAIL
         GO TO 20
      END IF
*      Generate the variates
      IFAIL = 1
      CALL G05SQF(N,A,B,STATE,X,IFAIL)
      IF (IFAIL.NE.0) THEN
         WRITE (NOUT,99997) IFAIL
         GO TO 20
      END IF
*      Display the variates
      WRITE (NOUT,99999) (X(I),I=1,N)
*
      20 CONTINUE
*
99999  FORMAT (1X,F10.4)
99998  FORMAT (1X,' ** G05KFF returned with IFAIL = ',I5)
99997  FORMAT (1X,' ** G05SQF returned with IFAIL = ',I5)
      END

```

9.2 Program Data

None.

9.3 Program Results

G05SQF Example Program Results

```

0.2727
-0.7870
0.4921
0.5965
-0.7908

```
